ロレンツの微分方程式の線形計算

SHIMURA Masato
jcd02773@nifty.com
URL:http://homepage3.nifty.com/asagaya_avenue

(2013/12/07)

目次

1	**山下の連立差分方程式のスクリプト**	1
2	**ロレンツモデル**	3
3	C.Reiter **の方法**	6

はじめに

M.I.T の気象学者 E.N. ロレンツが 1963 年作成した気象モデルのコンパクトな微分方程式はロレンツのストレイジ · アトラクタと呼ばれ、カオス研究の先駆けとなった。
2013 年はロレンツの気象モデル提示されてから 50 年になる。このロレンツの微分方程式を線形計算で見通しよくしてみよう。

1　山下の連立差分方程式のスクリプト

この微分方程式を差分化し、アトラクタを描く山下紀幸氏の見事なスクリプトは JAPLA のライブラリに入っている。

1. 微分方程式と差分計算のスクリプト
   山下の連立差分方程式のスクリプトは堅牢である。これをパラメータや初期値をス

クリプトを書き換えなくとも試すことができるよう両項型に修正する。

$$\begin{cases} X' = & -\sigma X & +\sigma Y & \\ Y' = & -XZ & +\gamma X & -Y \\ Z' = & XY & -bZ & \end{cases}$$

```
lorenz=: 4 : 0
dt=. 0.005   NB. dt
's r b'=. x  NB. sigma  gamma b
'X0 Y0 Z0'=. y
NB. using only once /next XX,YY,ZZ
XX=.X0 + dt * s *(Y0-X0)
YY=.Y0 + dt * ((r*X0)-(Y0+X0*Z0))
ZZ=.Z0 + dt * ((X0*Y0)-b*Z0)
XX,YY,ZZ
)
```

2. パラメータと初期値
   - パラメータ
     ロレンツのオリジナルパラメータは $\sigma = 10, b = \frac{8}{3}\gamma = 28$ である。J では $\frac{8}{3}$ は 8r3 とあらわす
   - *dt*
     引数が多すぎるので *dt* は 0.005 で固定したがここを変えると反復数で調整することとなる。
   - 初期値
     init は色々試してみよう。カオスは初期値如何に拘わらず挙動が似てくる。

```
init=: 5 8 10 NB. 初期値
```

3. Usage:

```
'noaxes' plot {|:  (10 8r3 28)&lorenz ^:(i.10000) init
```

`^:(i.10000)` は tacit 型の反復計算を 10000 回行う

4. plot 3D



2 ロレンツモデル

ロレンツは下から暖められ、上で冷える大気の流体モデルを考案し、無限個の変数を含む微分方程式体系を大胆に 3 個の変数以外は定数であるとするモデル単純化を行った。一つは滞留運動の速度 (x) で, 他の 2 は水平方向と垂直方向の温度変異 (y と z) を測るものとなった。

$$\begin{cases} X' = & -\sigma X & +\sigma Y & \\ Y' = & -XZ & +\gamma X & -Y \\ Z' = & XY & -bZ & \end{cases}$$

[*1]

そしてこの 3 個のパラメータは σ がプランドル数、γ がレイリー数、b が系の物理サイズに係わるものである。

$\sigma = 10, b = \dfrac{3}{8}, \gamma = 28$ がロレンツのオリジナル・パラメータである。

2.1 カオス

パラメータ σ と b はオリジナルのまま用いられる。γ を変化させることにより、定常状態から対流を起こし、突然カオスが発生する。

[*1] ロレンツモデルは非圧縮条件、ナビエ・ストークス方程式、熱伝導方程式の 3 式から導かれる。導出過程は [松葉] に詳しい

次の 4 個の図はオリジナルパラメータでの挙動である



青 X' , 赤 Y', 緑 Z'



X' と Y'



X' と Z'



Y' と Z'

γ の挙動

γ が 1 を越えると対流が始まる　　　　　　やがてロール状の対流が生じ





γ が 18 付近で突然ポップ分岐を起こし不安定になる

```
plot {|:  (10 8r3 94r5)&lorenz ^:(i.10000) init
```



2.2　線形化

ロレンツの連立微分方程式は線形化すると次のような系になる。(Hirsh p.312)

$$Y' = \begin{pmatrix} -\sigma & \sigma & 0 \\ \gamma - z & -1 & -x \\ y & x & -b \end{pmatrix} Y$$

3　C.Reiter の方法

C.Reiter の「Fractal visualization and J 第 3 版」でマトリクスとルンゲクッタ法でこの微分方程式を計算しているのを見受けた。マトリクス計算の好例なのでフォローしてみよう。

1. run

```
0.002 (10 8r3 28) lz rk ^:(i.100) 0.1 0.2 0.3
```

2. C.Reiter のロレンツのマトリクス。横長のマトリクスを用いて巧妙に作成されており、

```
   M
_10 10    0 0  0
 28 _1    0 0 _1
  0  0 _8r3 1  0
```

$$\left\{ \begin{array}{llll} X' = & -\sigma X & +\sigma Y & \\ Y' = & \gamma X & -Y & -XZ \\ Z' = & XY & & -bZ \end{array} \right.$$

$$Y' = \begin{pmatrix} -\sigma & \sigma & 0 \\ \gamma - z & -1 & -x \\ y & x & -b \end{pmatrix} Y$$

	x	y	z	xy	xz
X'	-10	10	0	0	0
Y'	28	-1	0	0	-1
Z'	0	0	$-8r3$	1	0

$Y' = \gamma X - Y - XZ$ の $-XZ$ と $Z' = XY - bZ$ の XY を別項に書き出している。ネスティッド・アレー（2 重配列）や文字列操作なども不要で、非線形方程式も横長マトリクスで計算できる。

3. 初期値とリピート入力　x, y, z と xy, xz の作成

```
   ([ , {. * }.) 0.1 0.2 0.3
0.1 0.2 0.3 0.02 0.03
```

4. C.Reiter の lz の script

```
NB. Lorenz
lz=: 1 : 0
'S B R'=. m                                          NB. s b r
M=. ((-S),S,0 0 0),(R,_1 0 0 _1),: 0 0 ,(-B),1 0   NB. make matrix
M&(+/ . *)@([ , {. * }.)                             NB. 計算部分
)
```

（a） `+/ . *` は内積計算を行う。`M (+/ . *)` **初期値**
（b）初期値の数はマトリクス M の列の数と等しくなければならない
（c）出力は M の行数すなわち連立方程式の元の数となる。
（d）ループ `^:(i. n)` では (c) の出力を 1 回づつ受けて、xz, yz をその都度構成して入力する

5. C.Reiter によるルンゲクッタ法のスクリプト。ループは外付けなので、実にシンプルである。

```
   rk=:1 : 0
:
h2=. -: x
k1=. u y
k2=. u y + h2*k1
k3=. u y + h2*k2
k4=. u y + x *k3
y+(x%6)*k1+k4++:k2+k3
)
```

- 1 : 0 は動詞 (関数) を引数に取る副詞型
- u は左の動詞を引数入力
- +: double= ×2

6. Usage:

```
'noaxes' plot {|:   0.002 (10 8r3 28 lz) rk ^:(i.10000) 0.1 0.2 0.3
```



References

Cliff Reiter [Fractal Visualization and J] 3rd edition Lulu 2007

M.Hirsch S.Smale R.L.Devaney 「力学系入門　原著　第 2 版」共立出版　2007

松葉育雄　「複雑系の数理」　朝倉書店　2004